

# FMX : Recherche dans un TListView

## Utilisation de l'évènement OnFilter

Par Serge Girard  

Date de publication : 8 février 2020

CONFIRMÉ

Le composant **TListView** permet de rajouter très facilement une boîte de recherche par l'intermédiaire de sa propriété **SearchVisible**. Sans autre ligne de code la recherche s'effectue sur tous les textes (même invisibles) de la liste. Toutefois cette recherche est établie selon le principe : « accepter l'affichage si un des textes contient le filtre sans tenir compte de la casse ». L'évènement **OnFilter** permet de modifier ce principe.

Mon objectif est de montrer comment utiliser cet évènement, d'en démontrer le mécanisme et de fournir des moyens supplémentaires pour des recherches encore plus ciblées.

I - Bases de l'utilisation de TListView.....	3
I-A - Les différentes apparences.....	3
I-B - Apparence personnelle (custom).....	4
I-C - L'apparence dynamique.....	4
I-D - Le remplissage avec les LiveBindings.....	5
I-E - La boîte de recherche.....	6
II - L'événement OnFilter.....	10
II-A - Cas d'utilisation simple, une seule zone texte.....	10
II-B - Cas d'utilisation avec 2 zones texte.....	11
II-B-1 - Mécanisme de l'évènement.....	11
II-B-2 - Ébauche de solution.....	12
II-C - Nombre d'objets de type texte d'un élément de liste.....	13
II-D - Rechercher sur une zone particulière.....	15
III - Rendre plus « générique » la méthode.....	16
III-A - Première ébauche.....	17
III-A-1 - Utilisation.....	19
III-B - Le coin de l' « expert ».....	20
III-B-1 - Utilisation.....	24
IV - Conclusion.....	24

## I - Bases de l'utilisation de TListView

On trouve beaucoup de vidéos et aide sur ce composant mais souvent sans s'attarder sur une utilisation plus poussée. Pour repousser ces manques, j'ai, moi aussi, beaucoup écrit à son propos dans d'autres tutoriels :

- **Mettre de la couleur dans un TListView ;**
- **Personnaliser un TListView : ajouter des pieds de groupes ;**
- TListView apparence dynamique et images.



Et ai communiqué sur mon blog :

- **Obtenir une liste groupée avec les Livebindings ;**
- **Obtenir une liste groupée avec les Livebindings. Suite ;**
- **Obtenir une liste groupée avec les Livebindings. Finalisation ?**
- **ListView FMX, entêtes et pieds de groupes : Suite et Fin !**
- **Un problème de ListView avec application de style enfin résolu ;**
- **Modifier la hauteur (et accessoirement d'autres propriétés) de la boîte de recherche d'une liste.**

Les différentes sections qui suivent vont surtout mettre en exergue ce qui pourrait influencer sur la recherche au sein des éléments de la liste.

**i** *Ce chapitre n'est qu'un survol de TListView. Vous trouverez une documentation relativement complète des possibilités de ce composant dans le **docwiki d'Embarcadero***

### I-A - Les différentes apparences



La plupart des apparences proposées ne contiennent que l'élément texte :

<b>TAppearanceNames</b> avec une seule zone texte
ListItem, ListItemDelete, ListItemShowCheck
ImageListItemRightButton, ImageListItemRightButtonDelete, ImageListItemRightButtonShowCheck
ImageListItem, ImageListItemDelete, ImageListItemShowCheck

Ou propose une zone détail en supplément :

<b>TAppearanceNames</b> avec une zone détail en supplément.
ImageListItemBottomDetail, ImageListItemBottomDetailShowCheck
ListItemRightDetail, ListItemRightDetailDelete, ListItemRightDetailShowCheck
ImageListItemBottomDetailRightButton, ImageListItemBottomDetailRightButtonShowCheck

## I-B - Apparence personnelle (custom)

Cette apparence prédéfinie est particulière, applicable également aux apparences d'entête (**HeaderAppearance**) et de pied (**FooterAppearance**) de groupe. Elle serait équivalente à l'apparence **ImageListItemBottomDetailRightButtonShowCheck** certains éléments étant non visibles par défaut.



À noter, la recherche ne s'effectuera jamais sur les entêtes et pied de groupes. Si, toutefois, vous vouliez faire une recherche sur le texte contenu dans ceux-ci la solution consistera à utiliser une apparence dynamique et d'ajouter un **TTextObjectAppearance**

## I-C - L'apparence dynamique

Apparue plus tardivement (Delphi Berlin 10.1), cette apparence permet un design beaucoup plus personnalisé et mettre, ainsi, plus de deux zones de texte.



Pour plus d'informations sur l'apparence dynamique, le plus simple est de commencer par lire ce que contient le Docwiki Embarcadero à ce sujet à [cette adresse](#).

Très succinctement, puisque ce n'est pas le sujet principal, après avoir sélectionné pour la propriété **ItemAppearance** la valeur **DynamicAppearance**, il est possible d'ajouter des objets spécifiques (ou plutôt des apparences d'objets) les uns à la suite des autres.


Cinq types d'objets sont possible :

<b>TTextObjectAppearance</b>	Un objet texte, le seul qui nous intéressera dans le cadre de ce tutoriel
<b>TImageObjectAppearance</b>	Un objet image
<b>TAccessoryObjectAppearance</b>	Un objet accessoire, trois types : <b>More</b> , <b>Checkmark</b> , <b>Detail</b>
<b>TTextButtonObjectAppearance</b>	Un objet bouton avec texte
<b>TGlyphButtonObjectAppearance</b>	Un objet bouton, trois types : <b>Add</b> , <b>Delete</b> , <b>CheckBox</b>



*Il est à noter que certains de ces objets ne sont visibles que si la liste est en mode édition. Le terme de ce mode est trompeur, vous ne pourrez pas pour autant, modifier les textes comme il serait possible de le faire dans une grille.*

Une fois les objets ajoutés, avec un peu d'adresse, il est possible d'arranger les tailles, positions et alignements de ceux-ci.



*Exemple d'apparence dynamique en conception*


## I-D - Le remplissage avec les LiveBindings

Qui dit TListView sous-entend le plus souvent un grand nombre de données, contrairement à un TListBox. L'éditeur de liaison nous permet de faire cela facilement.

Dans les projets accompagnateurs de ce tutoriel, j'utiliserai essentiellement un composant **TClientDataSet** pour avoir les données de deux fichiers situés dans le répertoire (**\$Samples**)\Data : **Biolife.xml** et **Customers.xml**.



*Sélectionner la liste et utiliser le menu contextuel (option : **lier visuellement**) pour réaliser les liens entre liste et données. Ce n'est l'affaire que d'une ou deux minutes pour obtenir une liste liée aux données si votre table est ouverte.*



Concepteur Livebindings projet RechercheBase

Cela ne veut pas dire que vous ne pourrez pas faire de recherche dans une liste remplie par code. Juste qu'il est très facile de remplir une liste sans code grâce aux LiveBindings.




Vous retrouverez également de nombreux tutoriels plus orientés Livebindings sur [mon site](#).

## I-E - La boîte de recherche



Pour illustrer ce chapitre, un projet nommé **RechercheBasique** : [téléchargeable ici](#)

Implémenter une boîte de recherche à un TListView est hyper facile puisqu'il suffit d'activer cette fonctionnalité en cochant la propriété **SearchVisible** de celle-ci.



Ajout boîte de recherche

Lorsqu'il n'y a qu'un seul élément la question de savoir sur quel texte sera faite une recherche ne se pose même pas. Par contre, lorsque ces éléments ont agrémentées d'un ou plus d'éléments contenant du texte, la recherche, par défaut, sera faite sur l'ensemble de ces éléments (visibles ou non).

Pratique dans la plupart des cas cet effet peut se révéler gênant dans certains cas.

Deux illustrations à ce propos :

- Dans ma liste de clients je veux faire une recherche par pays (Venezuela), taper seulement « **ven** » (vous noterez au passage que la recherche n'est pas sensible à la casse), me fourni un peu plus de résultats que prévu.

**i** *Taper « **vene** » me sors de l'embarras donc ce n'est pas flagrant. Mais, si je cache la zone pays, vous avouerez que cela deviendra étrange à l'utilisateur.*


Exemples de recherches		Exemples de recherches	
Customers	Biolife	Customers	Biolife
		Q ven	
ai Dive Shoppe	US	Adventure Undersea	B
aa Kauai		Belize City	
co	Bahamas	SCUBA Heaven	Baha
port		Nassau	
t Diver	Cyprus	Marina SCUBA Center	Venez
o Paphos		Caracas	
man Divers World Unlimited	British West Indies	Divers of Venice	
nd Cayman		Venice	
Sawyer Diving Centre	US Virgin Islands	Vashon Ventures	
istiansted		Honolulu	
Jack Aqua Center	US	Ocean Adventures	
oahu		Maui	
Divers Club	US Virgin Islands		
istiansted			

- Second exemple, j'affiche aussi la description de l'animal. Objectif avoué : obtenir tous les poissons dont le nom contient « **blu** ».



Exemples de recherches

Customers Biolife




Clown Triggerfish  
50

known as the big spotted triggerfish. Inhabits outer reef areas and s upon crustaceans and mollusks by crushing them with powerful n. They are voracious eaters, and divers report seeing the clown erfish devour beds of pearl oysters.

not eat this fish. According to an 1878 account, "the poisonous flesh primarily upon the nervous tissue of the stomach, occasioning violent ms of that organ, and shortly afterwards all the muscles of the body. frame becomes rocked with spasms, the tongue thickened, the eye d, the breathing laborious, and the patient expires in a paroxysm of me suffering."

edible.

ge is Indo-Pacific and East Africa to Somoa.




Red Emperor  
60

ed seaperch in Australia. Inhabits the areas around lagoon coral reefs sandy bottoms.

red emperor is a valuable food fish and considered a great sporting that fights with fury when hooked. The flesh of an old fish is just as

Exemples de recherches

Customers Biolife




Blue Angelfish  
30

Habitat is around boulders, caves, coral ledges and crevices in shallow waters. Swims alone or in groups.

Its color changes dramatically from juvenile to adult. The mature adult fish can startle divers by producing a powerful drumming or thumping sound intended to warn off predators.

Edibility is good.

Range is the entire Indo-Pacific region.



Cabezon  
99

Often called the great marbled sculpin. Found over rocky or shell-encrusted bottoms from shallow to moderately deep waters. It feeds primarily on crustaceans and mollusks.

The male cabezon will not budge while guarding its nest and can even be touched by divers.

Edibility is good; the flesh is bluish-green but turns white when cooked. The eggs of the cabezon are poisonous.

Range is from Alaska to Central Baja.

*Vous admettez que **Cabezon** ne rentre pas vraiment dans ces critères.*

**i** *Bien sûr, encore une fois, augmenter mes critères de recherches, rechercher « **blue** » fera disparaître **Cabezon** de la liste.*

En conclusion de cette petite démonstration (téléchargeable ici) ce qu'il nous faut, c'est un moyen de pouvoir affiner nos critères de recherche. Par exemple :

- Que celle-ci soit sensible à la casse ;
- Que celle-ci ne sélectionne l'élément que s'il commence par la valeur recherchée ;
- Que la sélection de soit faite que sur quelques éléments mais pas tous ;
- Et cœtera.

C'est là que va intervenir la codification dans l'évènement **OnFilter** de **TListView**.



Pour ne pas déroger au **principe de Pareto** (loi des 80-20) dans au moins 80 % des cas où vous aurez mis une possibilité de recherche dans une liste il n'y aura pas besoin de personnaliser cette recherche. La suite de ce tutoriel est de tenter de répondre aux 20 % restants.

## II - L'événement OnFilter

Note au lecteur :



À partir de maintenant, une fois établi que la recherche ne se fait que dans le texte contenus dans les éléments de type **TTextObjectAppearance**, dans le chapitre qui suit, **Objet**, **Objet Texte** en seront des synonymes.

Intéressons-nous aux paramètres de cet événement

Nom	Type	
Sender	TObject	Cela va de soi, il s'agit de l'objet à qui appartient l'évènement, donc un TListView
<b>const</b> AFilter	String	Le texte à rechercher
<b>const</b> AValue	String	Le point délicat ! D'où vient cette valeur et quid en cas de multiples objets contenant du texte ?
<b>var</b> Accept	Boolean	Variable qui va indiquer si l'on accepte que l'élément rempli la condition demandée

### II-A - Cas d'utilisation simple, une seule zone texte


Pour illustrer l'utilisation de l'évènement, intéressons-nous à une recherche de début de texte.

```
procedure TForm1.ListView1Filter(Sender: TObject; const AFilter, AValue: string; var
  Accept: Boolean);
begin
  Accept:=AFilter.IsEmpty OR AValue.StartsWith(AFilter, false);
  // le second paramètre de StartsWith permet d'indiquer la sensibilité
  // à la casse
end;
```

Simplissime non ?



Vous retrouverez ce code dans le projet **RechercheOnFilter** [téléchargeable ici](#)




☐ Recherche Normale    ☒ Recherche OnFilter  
☒ Sensible casse    ☒ Début texte

Kauai Dive Shoppe
Kirk Enterprises

*OnFilter un seul objet*

## II-B - Cas d'utilisation avec 2 zones texte

Prenons maintenant le cas d'un affichage de type **ImageListItemBottomDetail**



Recherche sur :

☒ Les deux objets    ☐ Compagnie    ☐ Pays

Kauai Dive Shoppe	US
Unisco	Bahamas
Sight Diver	Cyprus
Cayman Divers World Unlimited	British West Indies
Tom Sawyer Diving Centre	US Virgin Islands
Blue Jack Aqua Center	US
VIP Divers Club	US Virgin Islands
Ocean Paradise	US

Le problème principal est de déterminer comment est fourni **AValue**.

### II-B-1 - Mécanisme de l'évènement

Pour étudier ce mécanisme, il suffit de poser un point d'arrêt au sein de la fonction et de contrôler le contenu de la constante.

```

- | procedure TMainForm.LV2ObjetsFilter(Sender: TObject; const AFilter,
70 | AValue: string; var Accept: Boolean);
- | begin
- |   if AFilter.IsEmpty then Exit;
73 | Accept:= AValue.StartsWith(AFilter)
- | end;

```

Première constatation, l'évènement est fréquemment levé, en fait à chaque dessin d'un objet texte. Ainsi, successivement, le contenu de la constante **AValue** sera-t-il la valeur de l'objet **text** puis de l'objet **detail**.



*Successivement ? Non, pas tout à fait, si le premier objet testé satisfait la condition alors le second objet n'est pas testé et le programme passe tout de suite à l'élément de liste suivant.*

## II-B-2 - Ébauche de solution

A priori, il n'y a aucun moyen de savoir quel objet est en cours de dessin (du moins de manière simple). Introduire une sorte de compteur d'objets testés fut donc ma première approche.

### Ajout compteur

```

var
  MainForm: TMainForm;
  ItemObjectNumber : word;

```

### Initialisation compteur

```

initialization
  ItemObjectNumber:=0;
finalization
end.

```

Reste à charge de coder l'incrémement et la réinitialisation de ce compteur au sein de l'évènement **OnFilter**.


### OnFilter

```

procedure TMainForm.LV2ObjetsFilter(Sender: TObject; const AFilter,
  AValue: string; var Accept: Boolean);
begin
  if AFilter.IsEmpty then Exit; // pas de recherche
  inc(ItemObjectNumber); // incrémement
  // tests en fonction du mode
  if RBTexte.IsChecked then Accept := (ItemObjectNumber=1) AND
    UpperCase(AValue).Contains(UpperCase(AFilter));
  if RBDetail.IsChecked then Accept := (ItemObjectNumber=2) AND
    UpperCase(AValue).Contains(UpperCase(AFilter));
  // réinitialisation si le nombre d'objets texte est atteint
  // ou si la condition est réalisée (Accept:=true)
  if (ItemObjectNumber=2) OR Accept then ItemObjectNumber:=0;
end;

```

Dans ce contexte (toute apparence à base de **ListItemDetail** ou plus exactement ne contenant que deux objets de type texte) cette solution est fonctionnelle.



Elle peut même être applicable à de plus grand nombre d'objets de type texte, pour peu de bien maîtriser les diverses valeurs de la variable **ItemObjectNumber**.

En « puriste » je n'y vois que quelques bémols :

- 1 Le fait d'avoir à renseigner le nombre d'objets maximum **ItemObjectNumber=2**
- 2 Connaître l'ordre dans lequel seront affichés les objets de l'élément **Accept := (ItemObjectNumber=1) AND ...**
- 3 Le fait d'être obligé d'utiliser un variable globale (le compteur d'objets).

## II-C - Nombre d'objets de type texte d'un élément de liste

Le premier bémol peut, avec un peu d'expertise, être contourné en introduisant une variable privée (voire une propriété) calculée avant une quelconque opération de recherche sur la liste.

Pour les apparences dites « classiques », cf. **I.A**, le nombre d'objets est déductible facilement en recherchant dans le type d'apparence ( **<liste>.ItemAppearance.ItemAppearance** ) s'il contient le texte « **detail** » ou non.

```
nombredobjets:=1;
if LowerCase(aList.ItemAppearance.ItemAppearance).Contains('detail')
then Inc(nombredobjets);
```

Pour une apparence dynamique il va falloir vérifier au sein de la collection d'objets associée.

```
nombredobjets:=0;
if aList.ItemAppearance.ItemAppearance='DynamicAppearance'
then begin // apparence dynamique
  DynApp:=TDynamicAppearance(aList.ItemAppearanceObjects.ItemObjects);
  for appObj in DynApp.ObjectsCollection do
  begin
    if (appobj is TAppearanceObjectItem) AND
      (TAppearanceObjectItem(AppObj).Appearance is TTextObjectAppearance)
    then inc(nombredobjets);
  end;
end ;
```



Pour pouvoir accéder à la collection d'objets il est impératif d'ajouter dans les clauses d'utilisation (**uses**) l'unité **FMX.ListView.DynamicAppearance**

Le tout peut-être intégré dans une fonction plus générale :

#### Nombre d'objets texte

```
uses FMX.ListView.DynamicAppearance;
...

function TMainForm.HowManyTextObjects(const aList: TListView): integer;
var DynApp : TDynamicAppearance;
    appObj : TCollectionItem;
begin
  result:=0;
  if aList.ItemAppearance.ItemAppearance='DynamicAppearance'
  then begin
    DynApp:=TDynamicAppearance(aList.ItemAppearanceObjects.ItemObjects);
    for appObj in DynApp.ObjectsCollection do
    begin
      if (appobj is TAppearanceObjectItem) AND
        (TAppearanceObjectItem(AppObj).Appearance is TTextObjectAppearance)
      then inc(result);
    end;
  end
  else begin
    result:=1;
    if LowerCase(aList.ItemAppearance.ItemAppearance).Contains('detail')
    then Inc(result);
  end;
end;
```

ListItem	ListItemDetail	Dynamique	Tests
----------	----------------	-----------	-------

ListItem contient 1 objet


ListItemDetail contient 2 objets

Combien d'objets texte dans la liste Dynamique

## II-D - Rechercher sur une zone particulière

Par zone, j'entends selon le nom de l'objet texte.

*D'où l'importance de nommer ces objets plutôt que de laisser les noms par défaut : **Text1...** **Textn**, mais surtout faites le avant d'établir les liaisons !*



*Je n'ai pas encore trouvé de solution pour utiliser les noms de colonnes de la table liée. Même si cela doit être possible, je gage que cela compliquerait énormément le code.*

Une fois l'astuce d'obtention des objets d'une liste d'apparence dynamique connue il est simple de retrouver les noms des objets de la collection, la propriété **AppearanceObjectName**.

```
procedure GetTextObjectNames(aList : TListView);
var DynApp : TDynamicAppearance;
    appObj : TCollectionItem;
begin
    if aList.ItemAppearance.ItemAppearance='DynamicAppearance'
    then begin
        DynApp:=TDynamicAppearance(AList.ItemAppearanceObjects.ItemObjects);
        for appObj in DynApp.ObjectsCollection do
            begin
                if appObj is TAppearanceObjectItem then
                    if TAppearanceObjectItem(AppObj).Appearance is TTextObjectAppearance
                    then begin
                        Memo1.Lines.Add(TAppearanceObjectItem(AppObj).AppearanceObjectName);
                    end;
                end;
            end
        else begin
            Memo1.Lines.Add('text');
            if LowerCase(aList.ItemAppearance.ItemAppearance).Contains('detail')
```

```
then Memo1.Lines.Add('detail');
end;
end;
```

La liste Dynamique contient 3 objets

Noms des objets texte

☐ ListItem
 ☐ ListItemDetail
 ☒ Liste Dynamique

Client  
Ville  
Pays

Pour peu de stocker ces noms des zones qui doivent être filtrées dans une liste, j'ai ainsi la réponse à mon deuxième bémol du chapitre [II.B.2](#) en remplaçant

Accept := (ItemObjectNumber=1) AND ... par une instruction qui recherchera si l'objet en cours fait partie de cette liste ListeDesZonesATester.Contains(nomdelazoneencours) ;

*Encore un peu nébuleux ? Tout va se dévoiler dans le prochain chapitre.*



*Passez quand même un peu de temps sur ce second projet exemple, [téléchargeable ici](#), pour voir le comportement des recherches.*

### III - Rendre plus « générique » la méthode

Pourquoi rendre plus générique la méthode ? Tout simplement parce qu'ainsi stockées dans une unité à part elle pourra être utilisée dans d'autres formes d'une application ou, même, d'autres applications.

Reprenons les besoins, il nous faut :

- 1 Pouvoir spécifier un mode de recherche. Par défaut le texte de recherche doit être contenu (**Contains**). Dans le chapitre [II.A](#) j'ai fait la recherche sur le début de valeur (**StartsWith**). En plus de cette possibilité d'autres fonctions sont envisageables comme l'égalité (=) ou la recherche en fin de valeur (**EndsWith**).
- 2 Il faut également pouvoir indiquer si la recherche sera sensible ou non à la casse.
- 3 Intégrer le compteur d'objet tel que j'ai pu le définir au chapitre [II.B](#)
- 4 Faire en sorte d'obtenir le nombre d'objets de type texte cf. chapitre [II.C](#)
- 5 Enfin, la possibilité de rechercher sur une ou plusieurs zones de type texte doit faire partie de l'arsenal. Pour cela j'aurais besoin de deux listes, une contenant tout les noms d'objets de type texte d'un élément de liste, l'autre pour contenir le nom des différents objets que l'on veut prendre en compte (chapitre [II.D](#)).



## III-A - Première ébauche

Ma première approche a été de créer une nouvelle classe que je nommerai **TSearchInList** pour surclasser la recherche originelle.

Pour répondre au point numéro 1, je vais utiliser une énumération **TSearchMode**.

### unit SearchInListView

```
type
    TSearchMode = (smContains, smStartwith, smEndwith, smEqual);
```

Ma nouvelle classe sera composée de deux parties :

- Une partie privée.

Membres	Types	
Parent	TListView	Le parent de la classe, la liste appellante.
Fields	TList<String>	Liste des objets texte
CurIndice	SmallInt	Index de l'objet texte en cours
MaxIndice	SmallInt	Nombre des objets texte dans l'élément de liste

- Une partie publique.

Membres	Types	
Mode	TSearchMode	Le mode de recherche
CaseSensitive	Boolean	La sensibilité à la casse
TestFields	TList<String>	La liste des objets (noms) à tester
Constructor	-	Le constructeur de la classe
Destructor	-	Le destructeur de cette classe
Accept	function	Fonction à deux arguments AValue, AFilter renvoyant le résultat du test.

### unit SearchInListView

```
unit SearchInListView;

interface
uses System.SysUtils, System.Types, System.Classes, System.Math,
    System.Generics.Collections, System.Generics.Defaults,
    FMX.ListView, FMX.ListView.Types, FMX.ListView.Appearances;

type
    TSearchMode = (smContains, smStartwith, smEndwith, smEqual);

    TSearchinList = class(TObject)
    strict private
        Parent : TListView;
        TextObjectsNames : TList<String>; // Liste des noms des objets
        CurIndice : smallint; // numéro de l'objet en cours
    public
```

## unit SearchInListView

```

  Objects2Test : TList<String>; // liste des noms d'objets à tester
  Mode : TSearchMode;
  CaseSensitive : Boolean;
  constructor Create(AOwner : TListView);
  destructor Destroy; override;
  // ce qui va servir de filtre
  function Accept(const AFilter, AValue: string) : Boolean;
end;

implementation

uses FMX.ListView.DynamicAppearance;

function TSearchinList.Accept(const AFilter, AValue: string): Boolean;
var AVal, AFil, test : String;
begin
  Result:=True;
  if AFilter.IsEmpty then Exit;

  AVal:=AValue;
  AFil:=AFilter;
  if not casesensitive then
  begin
    AVal:=LowerCase(AValue);
    AFil:=LowerCase(AFilter);
  end;
  case Self.Mode of
    smContains : Result:=(Objects2Test.Count=0)
                  OR (Objects2Test.IndexOf(TextObjectsNames[CurIndice])>=0)
                  AND AVal.Contains(AFil);
    smStartwith : Result:=(Objects2Test.Count=0)
                  OR (Objects2Test.IndexOf(TextObjectsNames[CurIndice])>=0)
                  AND AVal.StartsWith(AFil);
    smEndwith : Result:=(Objects2Test.Count=0)
                  OR (Objects2Test.IndexOf(TextObjectsNames[CurIndice])>=0)
                  AND AVal.EndsWith(AFil);
    smEqual : Result:=(Objects2Test.Count=0)
                  OR (Objects2Test.IndexOf(TextObjectsNames[CurIndice])>=0)
                  AND (AVal=AFil);
  end;
  // position dans l'élément
  if Result then CurIndice:=0
  else inc(CurIndice);
  if CurIndice=TextObjectsNames.Count then CurIndice:=0;
end;

constructor TSearchinList.Create(AOwner: TListView);
var DynApp : TDynamicAppearance;
  appObj : TCollectionItem;
begin
  inherited Create;
  Parent:=AOwner;
  CurIndice:=0;
  MaxIndice:=-1; // nombre d'éléments
  Mode:=smContains;
  TextObjectsNames:=TList<String>.Create;
  // TestFields sera une liste insensible à la casse
  Objects2Test:=TList<String>.Create(TComparer<String>.Construct(
    function(const s1, s2: String): Integer
    begin
      Result := CompareText(s1, s2) ;
    end));
  // Remplissage de la liste des noms des TTextObjectAppearance
  if Parent.ItemAppearance.ItemAppearance='DynamicAppearance'
  then begin
    DynApp:=TDynamicAppearance(Parent.ItemAppearanceObjects.ItemObjects);
    for appObj in DynApp.ObjectsCollection do
    begin
      if appobj is TAppearanceObjectItem then
        if TAppearanceObjectItem(AppObj).Appearance is TTextObjectAppearance

```

### unit SearchInListView

```

    then
      TextObjectsNames.Add(TAppearanceObjectItem(AppObj).AppearanceObjectName);
    end;
  end
else begin
  TextObjectsNames.Add('text');
  if LowerCase(Parent.ItemAppearance.ItemAppearance).Contains('detail')
  then TextObjectsNames.Add('detail');
  end;
CaseSensitive:=False;
end;

destructor TSearchinList.Destroy;
begin
  TextObjectsNames.Free;
  Objects2Test.Free;
  inherited Destroy;
end;

end.
```

## III-A-1 - Utilisation

- 1 Ajouter l'unité **SearchInListView** à la listes des unités utilisées (partie interface).

```

interface

uses
  ...
  SearchInListView;
```

- 2 Déclarer une variable , privée ou publique, de type **TSearchInList**.

```

private
  { Déclarations privées }
  Search : TSearchInList;
```

- 3 Créer la variable lors de l'évènement **OnCreate** de la forme.

```

procedure TForm1.FormCreate(Sender: TObject);
begin
  Search:=TSearchInList.Create(ListView1);
  // Case Sensitive False; par défaut
end;
```

- 4 Codifier l'évènement **OnFilter** de la liste.

```

procedure TForm1.ListViewFilter(Sender: TObject; const AFilter,
  AValue: string; var Accept: Boolean);
begin
  Accept:=Search.Accept(AFilter,AValue);
end;
```

C'est prêt, il ne reste plus qu'à fournir à l'objet les propriétés en fonction de la demande

Exemples :

- Changer la sensibilité à la casse

```
Search.CaseSensitive:=chkCasse.IsChecked;
```

- Changer les zones de recherches

```

Search.TestFields.Clear;
Search.TestFields.Add('company');
```

```
Search.TestFields.Add('country');
```

- Changer le mode de recherche

```
Search.Mode:=TSearchMode.smStartwith;
```

Recherche sur :	Options
<input type="radio"/> Tous	<input type="radio"/> Compagnie
<input type="radio"/> Ville	<input checked="" type="radio"/> Pays
<input checked="" type="checkbox"/> Sensible à la casse	
<input type="radio"/> Contenant	<input type="radio"/> Débutant par
<input checked="" type="radio"/> Egal à	<input type="radio"/> se Terminant par

TabItem1	TabItem2
Q US	X
Kauai Dive Shoppe	US
Kapaa Kauai	
Blue Jack Aqua Center	US
Waipahu	
Ocean Paradise	US
Kailua-Kona	
The Depth Charge	US
Marathon	
Blue Sports	US
Giribaldi	
Makai SCUBA Club	US
Kailua-Kona	
Action Club	US
Sarasota	
Island Finders	US
St Simons Isle	
Blue Sports Club	US
Largo	



Vous retrouverez l'application de cette unité dans le projet *ListViewGenericSearch1* partie d'un ensemble de sources [téléchargeables ici](#)

### III-B - Le coin de l' « expert »

Le seul inconvénient, encore que minime, à la solution du chapitre [III.A](#) c'est qu'il va falloir déclarer une variable de type **TSearchInList** pour chacune des listes d'une unité pour peu, bien sûr, que nous ayons plusieurs **TListView** sur une même forme et que nous voulions implémenter cette fonctionnalité pour chacun de ces composants.

C'est en reprenant mes divers exemples que j'ai constaté ce léger problème. Mes autres recherches sur ce composant m'ont fait envisager une solution à base d'interface.

Pour ce faire, je suis reparti de la classe **TSearchInList** présentée chapitre **III.A**.

J'en ai extrait la partie interface au sens strict du terme.

#### SearchListViewInterface

```
unit SearchListViewInterface;

interface

uses System.SysUtils, System.Types, System.Classes,
    System.Generics.Collections;

type
  TSearchMode = (smContains, smStartwith, smEndwith, smEqual);

  ISearchInListView = interface
    ['{0AF62378-366E-4AED-9918-28F8BA5859A8}']

    function Accept(const AFilter, AValue: string): Boolean;
    function GetTextObjectsNames : TList<String>;
    // procedure SetTextObjectsNames(Value : TList<String>);
    function GetTestFields: TList<String>;
    procedure SetTestFields(Value: TList<String>);
    function GetMode: TSearchMode;
    procedure SetMode(Value: TSearchMode);
    function GetCaseSensitive: Boolean;
    procedure SetCaseSensitive(Value: Boolean);

    property TextObjectsNames : TList<String> read GetTextObjectsNames; // write
    SetTextObjectsNames;
    property TestObjectsTextName : TList<String> read GetTestFields write SetTestFields;
    property Mode: TSearchMode read GetMode write SetMode;
    property CaseSensitive: Boolean read GetCaseSensitive write SetCaseSensitive;
  end;

implementation

end.
```

Puis j'ai créé une unité qui va être utilisée pour surclasser le composant **TListView**.

#### Adaptateur

```
unit ListViewSearchAdapter;

interface

uses System.SysUtils, System.Types, System.Classes,
    System.Generics.Collections, System.Generics.Defaults,
    FMX.ListView, SearchListViewInterface;

Type
  // surclassement de TListView, ajout de l'interface
  TListView = Class(FMX.ListView.TListView, ISearchInListView)
  strict private
    FCurrent: SmallInt;
    FObjectsNames : TList<String>;
  private
    FMode: TSearchMode;
    FCaseSensitive: Boolean;
    FObjects2Test : TList<String>;
  public
    // ajout des éléments de l'interface
    constructor Create(AOwner: TComponent); override;
    destructor Destroy; override;
    function Accept(const AFilter, AValue: string): Boolean;
    function GetTextObjectsNames : TList<String>;
    // procedure SetTextObjectsNames(Value : TList<String>);
    function GetTestFields: TList<String>;
```

## Adaptateur

```

procedure SetTestFields(Value: TList<String>);
function GetMode: TSearchMode;
procedure SetMode(Value: TSearchMode);
function GetCaseSensitive: Boolean;
procedure SetCaseSensitive(Value: Boolean);

property Objects2Test : TList<String> read GetTestFields write SetTestFields;
procedure InitObjectsName;

end;

implementation

uses FMX.ListView.Types,
      FMX.ListView.Appearances,
      FMX.ListView.DynamicAppearance;

{ TListView }

function TListView.Accept(const AFilter, AValue: string): Boolean;
var
  AVal, AFil : String;
begin
  // récupère le nom des objets TTextObjectAppearance dans la liste
  if FObjectsNames.Count=0 then InitObjectsName;

  Result := True;
  if AFilter.IsEmpty then Exit;

  AVal := AValue;
  AFil := AFilter;
  if not FCaseSensitive then
  begin
    AVal := LowerCase(AValue);
    AFil := LowerCase(AFilter);
  end;
  case FMode of
    smContains:
      Result := ((FObjects2Test.Count = 0) OR
FObjects2Test.Contains(FObjectsNames.Items[FCurrent]))
        AND AVal.Contains(AFil);
    smStartwith:
      Result := ((FObjects2Test.Count = 0) OR
FObjects2Test.Contains(FObjectsNames.Items[FCurrent]))
        AND AVal.StartsWith(AFil);
    smEndwith:
      Result := ((FObjects2Test.Count = 0) OR
FObjects2Test.Contains(FObjectsNames.Items[FCurrent]))
        AND AVal.EndsWith(AFil);
    smEqual:
      Result := ((FObjects2Test.Count = 0) OR
FObjects2Test.Contains(FObjectsNames.Items[FCurrent]))
        AND (AVal = AFil);
  end;
  if Result then
    FCurrent := 0
  else begin
    inc(FCurrent);
    if FCurrent = (FObjectsNames.Count) then FCurrent := 0;
  end;
end;

constructor TListView.Create(AOwner: TComponent);
begin
  inherited;
  FObjectsNames:=TList<String>.Create;
  FObjects2Test:=TList<String>.Create(TComparer<String>.Construct(
    function(const s1, s2: String): Integer
    begin
      Result := CompareText(s1, s2) ;
    end;
  ));
end;

```

## Adaptateur

```

    end);
    FCurrent:=0;
    FMode:=smContains;
    FCaseSensitive:=False;
end;

destructor TListView.Destroy;
begin
    FreeAndNil (FObjectsNames);
    FreeAndnil (FObjects2Test);
    inherited;
end;

function TListView.GetCaseSensitive: Boolean;
begin
    Result:=FCaseSensitive;
end;

function TListView.GetMode: TSearchMode;
begin
    Result:=FMode;
end;

function TListView.GetTextObjectsNames: TList<String>;
begin
    result:= FObjectsNames;
end;

function TListView.GetTestFields: TList<String>;
begin
    result := FObjects2Test;
end;

procedure TListView.SetCaseSensitive(Value: Boolean);
begin
    FCaseSensitive:=Value;
end;

procedure TListView.InitObjectsName;
var
    DynApp: TDynamicAppearance;
    appObj: TCollectionItem;
begin
    if not Assigned(FObjectsNames) then
        FObjectsNames := TList<String>.Create;
    // obtenir les éléments texte
    if ItemAppearance.ItemAppearance = 'DynamicAppearance' then
    begin
        if EditMode
        then DynApp := TDynamicAppearance(ItemAppearanceObjects.ItemEditObjects)
        else DynApp := TDynamicAppearance(ItemAppearanceObjects.ItemObjects);
        for appObj in DynApp.ObjectsCollection do
        begin
            if appObj is TAppearanceObjectItem then
                if TAppearanceObjectItem(appObj).Appearance is TTextObjectAppearance
                then
                    FObjectsNames.Add (TAppearanceObjectItem(appObj).AppearanceObjectName);
                end;
            end
        else
        begin
            FObjectsNames.Add('text');
            if LowerCase (ItemAppearance.ItemAppearance).Contains('detail')
            then FObjectsNames.Add('detail');
            end;
        end;
    end;

procedure TListView.SetMode(Value: TSearchMode);
begin
    FMode := Value;
end;

```

## Adaptateur

```
//procedure TListView.SetObjectsTextNames(Value: TList<String>);
//begin
//FObjectsName:=Value;
//end;

procedure TListView.SetTestFields(Value: TList<String>);
begin
    if not Assigned(FObjects2Test) then
        FObjects2Test := TList<string>.Create;
        FObjects2Test := Value;
end;

end.
```

## III-B-1 - Utilisation

### 1 Déclarer l'utilisation des unités

```
interface

uses
    ...
    FMX.ListView, SearchListViewInterface, ListViewSearchAdapter;
```



*Il est important de déclarer ces unités dans la partie interface, mais surtout, **après** l'utilisation de l'unité **FMX.ListView** pour que le surclassement opère.*

### 1 Dans le code

```
// Changement de sensibilité à la casse (exemples)
ListView1.SetCaseSensitive(True);
ListView1.SetCaseSensitive(False);

// Changement de mode (exemples)
ListView1.SetMode(TSearchMode.smContains);
ListView1.SetMode(TSearchMode.smStartwith);
ListView1.SetMode(TSearchMode.smEqual);
ListView1.SetMode(TSearchMode.smEndwith);

// Objets à tester (exemples)
ListView1.Objects2Test.Add('text');
ListView1.Objects2Test.Add('detail');
ListView1.Objects2Test.Add('country');
ListView1.Objects2Test.AddRange(['city', 'country']);

// Codification de l'évènement OnFilter de la liste
procedure TForm1.ListView1Filter(Sender: TObject; const AFilter,
    AValue: string; var Accept: Boolean);
begin
    // appel de la fonction Accept de l'interface
    Accept:=ListView1.Accept(AFilter,AValue);
end;
```

## IV - Conclusion

L'objectif de ce tutoriel était de démontrer les mécanismes de la recherche de TListView ainsi que les moyens de customiser celle-ci. Les différentes étapes permettant de comprendre la mise en œuvre et d'en apprendre un peu plus sur le surclassement et les interfaces.



Je le répète, dans la plupart des cas vous n'aurez pas besoin de surclasser ainsi votre TListView.

Perspectives :

- Il serait sympathique de pouvoir mettre en exergue le texte recherché au sein des éléments, en utilisant les propriétés de **TTextLayout** c'est envisageable (voir **chapitre VII.C de ce tutoriel**), si je réussis à le faire je ne manquerai pas de le communiquer sur mon blog.
- Bien que différent de TListView, il doit être possible d'utiliser la même technique pour un TListBox, peut-être le sujet d'un nouveau tutoriel.

Je remercie tous les intervenants qui ont pu m'aider à peaufiner chaque partie lors de mes appels à l'aide sur **le forum**. Ayez, comme moi, une pensée à l'équipe rédactionnelle, les correcteurs techniques comme grammaticaux sans qui ce tutoriel ne pourrait être sous vos yeux.